

=====
Program: BumpFunction.mw

Description: Derivation of the analytic solution to test cases using the bump function. The cases include:
1) a 2D circular scalar distribution
2) a 3D spherical scalar distribution
3) a 3D torus vector distribution

Publication: [1] 'A high order solver for the unbounded Poisson equation'
J. Comput. Phys.

Authors: Mads Mølholm Hejlesen (a)
Johannes Tophøj Rasmussen (a)
Philippe Chatelain (b)
Jens Honore Walther (a,c)*

(a) Technical University of Denmark

(b) Universite catholique de Louvain

(c) ETH Zürich

*Corresponding author at jhw@mek.dtu.dk
=====

THE BUMP FUNCTION

> restart: with(plots):

- Defining the bump function [1] Eq. (30)

> f := exp(-c/(1-rho^2));

$$f := e^{-\frac{c}{1-\rho^2}}$$

(1)

2D scalar distribution: VORTICITY PATCH

- Define the stream function [1] Eq. (31) using cartesian coordinates (x,y,z)

> psi := subs(rho = sqrt(x^2 + y^2)/R, f);

$$\psi := e^{-\frac{c}{1 - \frac{x^2 + y^2}{R^2}}}$$

(2)

- Calculate the vorticity by [1] Eq. (3)

```
> omega := simplify(-(diff(psi,x,x) + diff(psi,y,y)));
```

$$\omega := \frac{4 c e^{-\frac{c R^2}{R^2 - x^2 - y^2}} R^2 (-x^4 - 2 x^2 y^2 + R^4 - y^4 - c x^2 R^2 - c y^2 R^2)}{(R^2 - x^2 - y^2)^4} \quad (3)$$

- Calculate the velocity components by [1] Eq. (2)

```
> ux := diff(psi,y);
```

$$u_x := -\frac{2 c y e^{-\frac{c}{1 - \frac{x^2 + y^2}{R^2}}}}{\left(1 - \frac{x^2 + y^2}{R^2}\right)^2 R^2} \quad (4)$$

```
> uy := -diff(psi,x);
```

$$u_y := \frac{2 c x e^{-\frac{c}{1 - \frac{x^2 + y^2}{R^2}}}}{\left(1 - \frac{x^2 + y^2}{R^2}\right)^2 R^2} \quad (5)$$


- - -
- Plotting stream function(blue), vorticity (red) and z-velocity(green) along (x,y) = (:, 0)
- - -

```
> p1 := plot(subs(c = 20, R = 1, y = 0,psi)/subs(c = 20, R = 1, y = 0, x=0,psi), x= -1 .. 1, color = blue):
```

```
> p2 := plot(subs(c = 20, R = 1, y = 0,omega)/subs(c = 20, R = 1, y = 0, x=0,omega), x= -1 .. 1, color = red):
```

```
> p3 := plot(subs(c = 20, R = 1, y = 0,uy)/subs(c = 20, R = 1, y = 0, x=0.15,uy), x= -0.99 .. 0.99, color = green):
```

```
> display({p1,p2,p3});
```



3D scalar distribution: SPHERE

- Define the stream function using spherical coordinates (r,θ,φ)

> **psi := subs(rho = r/R,f);**

$$\psi := e^{-\frac{c}{1 - \frac{r^2}{R^2}}}$$

(6)

- Calculate the vorticity by [1] Eq. (3)

> **omega := simplify(-(1/r^2*diff(r^2*diff(psi,r),r)));**

$$\omega := \frac{2 c e^{-\frac{c R^2}{(R-r)(R+r)}} R^2 (3 R^4 - 2 R^2 r^2 - r^4 - 2 r^2 c R^2)}{(R^2 - r^2)^4}$$


(7)

- Plotting stream function(blue), vorticity (red) along (x,y,z) = (:,0,0)

```

> p1 := plot(subs(c = 20, R = 1, psi)/subs(c = 20, R = 1, r=0, psi),
  r= -1 .. 1, color = blue):
> p2 := plot(subs(c = 20, R = 1, omega)/subs(c = 20, R = 1, r=0,
  omega), r= -1 .. 1, color = red):
> display({p1,p2});

```



3D vector distribution: VORTEX RING (xy-plane)

- Define the stream function [1] Eq. (31) using cylindrical coordinates (r, θ, z)

```

> psi := subs(rho = sqrt((r-R)^2 + z^2)/R, f);

```

$$\psi := e^{-\frac{c}{1 - \frac{R^2 - 2Rr + r^2 + z^2}{R^2}}}$$

(8)

```

> psi1 := -sin(theta)*subs(rho = sqrt((r-R)^2 + z^2)/R, f);

```

$$\psi1 := -\sin(\theta) e^{-\frac{c}{1 - \frac{R^2 - 2Rr + r^2 + z^2}{R^2}}} \quad (9)$$

> psi2 := cos(theta)*subs(rho = sqrt((r-R)^2 + z^2)/R,f);

$$\psi2 := \cos(\theta) e^{-\frac{c}{1 - \frac{R^2 - 2Rr + r^2 + z^2}{R^2}}} \quad (10)$$

> psi3 := 0;

$$\psi3 := 0 \quad (11)$$

- Calculate the vorticity by [1] Eq. (3)

> omega1 := simplify(-(1/r*diff(r*diff(psi1,r),r) + 1/r^2*diff(psi1,theta,theta) + diff(psi1,z,z)));

$$\omega1 := \frac{1}{(2Rr - r^2 - z^2)^4 r^2} \left(\sin(\theta) e^{-\frac{cR^2}{2Rr - r^2 - z^2}} (4c^2 z^2 R^4 r^2 - 8r^3 c R^5 + 8r^4 c R^4 - 6r^5 c R^3 + 4r^2 c^2 R^6 - 8r^3 c^2 R^5 + 4r^4 c^2 R^4 + 2r^6 c R^2 + 32R^3 r^3 z^2 - 48R^2 r^4 z^2 - 24R^2 r^2 z^4 + 24Rr^5 z^2 + 24Rr^3 z^4 + 8Rr z^6 + 2rcR^3 z^4 - 4r^3 cR^3 z^2 + 2r^2 cR^2 z^4 + 4cz^2 R^2 r^4 - r^8 - z^8 - 16R^4 r^4 + 32R^3 r^5 - 24R^2 r^6 + 8Rr^7 - 4r^6 z^2 - 6r^4 z^4 - 4r^2 z^6) \right) \quad (12)$$

> omega2 := simplify(-(1/r*diff(r*diff(psi2,r),r) + 1/r^2*diff(psi2,theta,theta) + diff(psi2,z,z)));

$$\omega2 := -\frac{1}{(2Rr - r^2 - z^2)^4 r^2} \left(\cos(\theta) e^{-\frac{cR^2}{2Rr - r^2 - z^2}} (4c^2 z^2 R^4 r^2 - 8r^3 c R^5 + 8r^4 c R^4 - 6r^5 c R^3 + 4r^2 c^2 R^6 - 8r^3 c^2 R^5 + 4r^4 c^2 R^4 + 2r^6 c R^2 + 32R^3 r^3 z^2 - 48R^2 r^4 z^2 - 24R^2 r^2 z^4 + 24Rr^5 z^2 + 24Rr^3 z^4 + 8Rr z^6 + 2rcR^3 z^4 - 4r^3 cR^3 z^2 + 2r^2 cR^2 z^4 + 4cz^2 R^2 r^4 - r^8 - z^8 - 16R^4 r^4 + 32R^3 r^5 - 24R^2 r^6 + 8Rr^7 - 4r^6 z^2 - 6r^4 z^4 - 4r^2 z^6) \right) \quad (13)$$

> omega3 := simplify(-(1/r*diff(r*diff(psi3,r),r) + 1/r^2*diff(psi3,theta,theta) + diff(psi3,z,z)));

$$\omega3 := 0 \quad (14)$$

- Calculate the velocity components by [1] Eq. (2)

> ur := simplify(-diff(psi,z));

$$ur := \frac{2cze^{-\frac{cR^2}{2Rr - r^2 - z^2}} R^2}{(2Rr - r^2 - z^2)^2} \quad (15)$$

```
> utheta := 0;
                                     utheta:= 0
```

(16)

```
> uz := simplify(1/r*diff(r*psi,r));
uz:=
```

(17)

$$\frac{1}{r(2Rr-r^2-z^2)^2} \left(e^{-\frac{cR^2}{2Rr-r^2-z^2}} (4R^2r^2-4Rr^3-4Rrz^2+r^4+2r^2z^2+z^4+2rcR^3-2r^2cR^2) \right)$$

- Transform the velocity components to cartesian coordinates (x,y,z)

```
> ux := cos(theta)*ur;
```

$$ux:= \frac{2 \cos(\theta) c z e^{-\frac{cR^2}{2Rr-r^2-z^2}} R^2}{(2Rr-r^2-z^2)^2}$$

(18)

```
> uy := sin(theta)*ur;
```

$$uy:= \frac{2 \sin(\theta) c z e^{-\frac{cR^2}{2Rr-r^2-z^2}} R^2}{(2Rr-r^2-z^2)^2}$$

(19)

```
> uz := uz;
uz:=
```

(20)

$$\frac{1}{r(2Rr-r^2-z^2)^2} \left(e^{-\frac{cR^2}{2Rr-r^2-z^2}} (4R^2r^2-4Rr^3-4Rrz^2+r^4+2r^2z^2+z^4+2rcR^3-2r^2cR^2) \right)$$

- Plotting stream function(blue), vorticity (red) and z-velocity(green) along (x,y,z) = (:,0,0)

```
> p1a := plot(subs(c = 20, R = 1, theta = Pi, z = 0, r = abs(r),
psi2)/subs(c = 20, R = 1, theta = 0, z = 0, r = 1, psi2), r= -2 ..
0, color = blue);
> p1b := plot(subs(c = 20, R = 1, theta = 0, z = 0, r = abs(r),
psi2)/subs(c = 20, R = 1, theta = 0, z = 0, r = 1, psi2), r= 0 ..
2, color = blue);
> p2a := plot(subs(c = 20, R = 1, theta = Pi, z = 0, r = abs(r),
omega2)/subs(c = 20, R = 1, theta = 0, z = 0, r = 1, omega2), r=
-2 .. 0, color = red);
```

```

> p2b := plot(subs(c = 20, R = 1, theta = 0, z = 0, r = abs(r),
omega2)/subs(c = 20, R = 1, theta = 0, z = 0, r = 1, omega2), r=
0 .. 2, color = red):
> p3a := plot(subs(c = 20, R = 1, theta = Pi, z = 0, r = abs(r),
uz)/subs(c = 20, R = 1, theta = 0, z = 0, r = 0.85, uz), r= -2 ..
0, color = green):
> p3b := plot(subs(c = 20, R = 1, theta = 0, z = 0, r = abs(r), uz)
/subs(c = 20, R = 1, theta = 0, z = 0, r = 0.85, uz), r= 0 .. 2,
color = green):
> display({p1a,p1b,p2a,p2b,p3a,p3b});

```

