

=====
Program: IntegrationKernels2D.mw

Description: Derivation of 2D high order integration kernels.

Publication: [1] 'A high order solver for the unbounded Poisson equation'
J. Comput. Phys.

Authors: Mads Mølholm Hejlesen (a)
 Johannes Tophøj Rasmussen (a)
 Philippe Chatelain (b)
 Jens Honore Walther (a,c)*

(a) Technical University of Denmark

(b) Universite catholique de Louvain

(c) ETH Zürich

*Corresponding author at jhw@mek.dtu.dk
=====

> **restart; with(plots):**

- Define the Gauss function

> **gauss := exp(-rho^2/2);**

$$gauss := e^{-\frac{1}{2} \rho^2} \quad (1)$$

- Adding one degree of freedom to solve for the normalization

> **z02 := A*gauss;**

$$z02 := A e^{-\frac{1}{2} \rho^2} \quad (2)$$

> **s1 := 2*Pi*int(rho*z02, rho = 0 .. infinity);**

$$s1 := 2 \pi A \quad (3)$$

> **sol := solve({s1 = 1}, {A});**

$$sol := \left\{ A = \frac{1}{2 \pi} \right\} \quad (4)$$

- Defining the smoothing polynomialia using recursion [1] Eq. (25)

> **p02 := rhs(sol[1]);**

(5)

$$p02 := \frac{1}{2\pi} \quad (5)$$

> **p04 := simplify((1 + 2/2 - 1/2*rho^2)* p02 + 1/2 * rho * diff(p02,rho));**

$$p04 := \frac{1 - \frac{1}{4}\rho^2}{\pi} \quad (6)$$

> **p06 := simplify((1 + 2/4 - 1/4*rho^2)* p04 + 1/4 * rho * diff(p04,rho));**

$$p06 := \frac{1}{16} \frac{\rho^4 - 12\rho^2 + 24}{\pi} \quad (7)$$

> **p08 := simplify((1 + 2/6 - 1/6*rho^2)* p06 + 1/6 * rho * diff(p06,rho));**

$$p08 := -\frac{1}{96} \frac{\rho^6 - 24\rho^4 + 144\rho^2 - 192}{\pi} \quad (8)$$

> **p10 := simplify((1 + 2/8 - 1/8*rho^2)* p08 + 1/8 * rho * diff(p08,rho));**

$$p10 := \frac{1}{768} \frac{\rho^8 - 40\rho^6 + 480\rho^4 - 1920\rho^2 + 1920}{\pi} \quad (9)$$

- Calculating the weight function by [1] Eq. (21)

> **w02 := simplify(1 + 2*Pi*int(rho*p02*gauss,rho));**

$$w02 := 1 - e^{-\frac{1}{2}\rho^2} \quad (10)$$

> **w04 := simplify(1 + 2*Pi*int(rho*p04*gauss,rho));**

$$w04 := 1 + \frac{1}{2}\rho^2 e^{-\frac{1}{2}\rho^2} - e^{-\frac{1}{2}\rho^2} \quad (11)$$

> **w06 := simplify(1 + 2*Pi*int(rho*p06*gauss,rho));**

$$w06 := 1 - \frac{1}{8} e^{-\frac{1}{2}\rho^2} \rho^4 + \rho^2 e^{-\frac{1}{2}\rho^2} - e^{-\frac{1}{2}\rho^2} \quad (12)$$

> **w08 := simplify(1 + 2*Pi*int(rho*p08*gauss,rho));**

$$w08 := 1 + \frac{1}{48} e^{-\frac{1}{2}\rho^2} \rho^6 - \frac{3}{8} e^{-\frac{1}{2}\rho^2} \rho^4 + \frac{3}{2} \rho^2 e^{-\frac{1}{2}\rho^2} - e^{-\frac{1}{2}\rho^2} \quad (13)$$

> **w10 := simplify(1 + 2*Pi*int(rho*p10*gauss,rho));**

$$w10 := 1 - \frac{1}{384} e^{-\frac{1}{2}\rho^2} \rho^8 + \frac{1}{12} e^{-\frac{1}{2}\rho^2} \rho^6 - \frac{3}{4} e^{-\frac{1}{2}\rho^2} \rho^4 + 2\rho^2 e^{-\frac{1}{2}\rho^2} - e^{-\frac{1}{2}\rho^2} \quad (14)$$

- Calculating the integration kernels by [1] Eq. (22)

> **G02 := -int(w02/rho,rho)/(2*Pi);**

$$G02 := -\frac{1}{2} \frac{\ln(\rho) + \frac{1}{2} \text{Ei}\left(1, \frac{1}{2} \rho^2\right)}{\pi} \quad (15)$$

> **G04 := -int(w04/rho,rho)/(2*Pi);**

$$G04 := -\frac{1}{2} \frac{\ln(\rho) - \frac{1}{2} e^{-\frac{1}{2} \rho^2} + \frac{1}{2} \text{Ei}\left(1, \frac{1}{2} \rho^2\right)}{\pi} \quad (16)$$

> **G06 := -int(w06/rho,rho)/(2*Pi);**

$$G06 := -\frac{1}{2} \frac{\ln(\rho) + \frac{1}{8} \rho^2 e^{-\frac{1}{2} \rho^2} - \frac{3}{4} e^{-\frac{1}{2} \rho^2} + \frac{1}{2} \text{Ei}\left(1, \frac{1}{2} \rho^2\right)}{\pi} \quad (17)$$

> **G08 := -int(w08/rho,rho)/(2*Pi);**

$$G08 := -\frac{1}{2} \frac{\ln(\rho) - \frac{1}{48} e^{-\frac{1}{2} \rho^2} \rho^4 + \frac{7}{24} \rho^2 e^{-\frac{1}{2} \rho^2} - \frac{11}{12} e^{-\frac{1}{2} \rho^2} + \frac{1}{2} \text{Ei}\left(1, \frac{1}{2} \rho^2\right)}{\pi} \quad (18)$$

> **G10 := -int(w10/rho,rho)/(2*Pi);**

$$G10 := -\frac{1}{2} \frac{1}{\pi} \left(\ln(\rho) + \frac{1}{384} e^{-\frac{1}{2} \rho^2} \rho^6 - \frac{13}{192} e^{-\frac{1}{2} \rho^2} \rho^4 + \frac{23}{48} \rho^2 e^{-\frac{1}{2} \rho^2} - \frac{25}{24} e^{-\frac{1}{2} \rho^2} + \frac{1}{2} \text{Ei}\left(1, \frac{1}{2} \rho^2\right) \right) \quad (19)$$

- Calculating the velocity kernels by $K = \nabla G \times$

> **K02 := simplify(diff(G02,rho));**

$$K02 := \frac{1}{2} \frac{-1 + e^{-\frac{1}{2} \rho^2}}{\rho \pi} \quad (20)$$

> **K04 := simplify(diff(G04,rho));**

$$K04 := -\frac{1}{4} \frac{\rho^2 e^{-\frac{1}{2} \rho^2} - 2 e^{-\frac{1}{2} \rho^2} + 2}{\rho \pi} \quad (21)$$

> **K06 := simplify(diff(G06,rho));**

$$K06 := \frac{1}{16} \frac{e^{-\frac{1}{2} \rho^2} \rho^4 - 8 \rho^2 e^{-\frac{1}{2} \rho^2} + 8 e^{-\frac{1}{2} \rho^2} - 8}{\rho \pi} \quad (22)$$

> **K08 := simplify(diff(G08,rho));**


$$K08 := -\frac{1}{96} \frac{e^{-\frac{1}{2}\rho^2} \rho^6 - 18 e^{-\frac{1}{2}\rho^2} \rho^4 + 72 \rho^2 e^{-\frac{1}{2}\rho^2} - 48 e^{-\frac{1}{2}\rho^2} + 48}{\rho \pi} \quad (23)$$

```
> K10 := simplify(diff(G10, rho));
```

$$K10 := \frac{1}{768} \frac{e^{-\frac{1}{2}\rho^2} \rho^8 - 32 e^{-\frac{1}{2}\rho^2} \rho^6 + 288 e^{-\frac{1}{2}\rho^2} \rho^4 - 768 \rho^2 e^{-\frac{1}{2}\rho^2} + 384 e^{-\frac{1}{2}\rho^2} - 384}{\rho \pi} \quad (24)$$


 - Plotting the smoothing functions ζ (one-sided) [1] Eq. (23)

```
> p1 := plot(p02*gauss, rho = 0 .. 5, color = blue):
> p2 := plot(p04*gauss, rho = 0 .. 5, color = red):
> p3 := plot(p06*gauss, rho = 0 .. 5, color = green):
> p4 := plot(p08*gauss, rho = 0 .. 5, color = magenta):
> p5 := plot(p10*gauss, rho = 0 .. 5, color = cyan):
> display({p1,p2,p3,p4,p5});
```



 - Plotting the G kernels (one sided, black = non-regularised kernel)

```
> p1 := plot(-log(x)/(2*Pi), x = 0 .. 5, y = -0.3 .. 0.5, color =
  black):
> p2 := plot(G02, rho = 0 .. 5, color = blue):
> p3 := plot(G04, rho = 0 .. 5, color = red):
> p4 := plot(G06, rho = 0 .. 5, color = green):
> p5 := plot(G08, rho = 0 .. 5, color = magenta):
> p6 := plot(G10, rho = 0 .. 5, color = cyan):
> display({p1,p2,p3,p4,p5,p6});
```



- Plotting the K kernels (one-sided, black = non-regularised kernel)

```
> p1 := plot(x/(2*Pi*x^2), x = 0 .. 5, y = 0 .. 0.5, color = black)
:
> p2 := plot(-K02, rho = 0 .. 5, color = blue):
> p3 := plot(-K04, rho = 0 .. 5, color = red):
> p4 := plot(-K06, rho = 0 .. 5, color = green):
> p5 := plot(-K08, rho = 0 .. 5, color = magenta):
> p6 := plot(-K10, rho = 0 .. 5, color = cyan):
> display({p1,p2,p3,p4,p5,p6});
```

