

=====
Program: IntegrationKernels3D.mw

Description: Derivation of 3D high order integration kernels.

Publication: [1] 'A high order solver for the unbounded Poisson equation'
J. Comput. Phys.

Authors: Mads Mølholm Hejlesen (a)
 Johannes Tophøj Rasmussen (a)
 Philippe Chatelain (b)
 Jens Honore Walther (a,c)*

(a) Technical University of Denmark

(b) Universite catholique de Louvain

(c) ETH Zürich

*Corresponding author at jhw@mek.dtu.dk
=====

> **restart; with(plots):**

- Define the Gauss function

> **gauss := exp(-rho^2/2);**

$$gauss := e^{-\frac{1}{2}\rho^2}$$

(1)

- Adding one degree of freedom to solve for the normalization

> **z02 := A*gauss;**

$$z02 := A e^{-\frac{1}{2}\rho^2}$$

(2)

> **s1 := 4*Pi*(int(rho^2*z02, rho = 0 .. infinity));**

$$s1 := 2 \pi^{3/2} A \sqrt{2}$$

(3)

> **sol := solve({s1 = 1}, {A});**

$$sol := \left\{ A = \frac{1}{4} \frac{\sqrt{2}}{\pi^{3/2}} \right\}$$

(4)

- Defining the smoothing polynomia using recursion [1] Eq. (25)

> **p02 := rhs(sol[1]);**

$$p02 := \frac{1}{4} \frac{\sqrt{2}}{\pi^{3/2}} \quad (5)$$

> p04 := simplify((1 + 3/2 - 1/2*rho^2)* p02 + 1/2 * rho * diff(p02,rho));

$$p04 := -\frac{1}{8} \frac{(\rho^2 - 5) \sqrt{2}}{\pi^{3/2}} \quad (6)$$

> p06 := simplify((1 + 3/4 - 1/4*rho^2)* p04 + 1/4 * rho * diff(p04,rho));

$$p06 := \frac{1}{32} \frac{\sqrt{2} (\rho^4 - 14\rho^2 + 35)}{\pi^{3/2}} \quad (7)$$

> p08 := simplify((1 + 3/6 - 1/6*rho^2)* p06 + 1/6 * rho * diff(p06,rho));

$$p08 := -\frac{1}{192} \frac{\sqrt{2} (\rho^6 - 27\rho^4 + 189\rho^2 - 315)}{\pi^{3/2}} \quad (8)$$

> p10 := simplify((1 + 3/8 - 1/8*rho^2)* p08 + 1/8 * rho * diff(p08,rho));

$$p10 := \frac{1}{1536} \frac{\sqrt{2} (\rho^8 - 44\rho^6 + 594\rho^4 - 2772\rho^2 + 3465)}{\pi^{3/2}} \quad (9)$$

- Calculating the weight function by [1] Eq. (21)

> w02 := simplify(4*Pi*int(rho^2*p02*gauss,rho));

$$w02 := \frac{1}{2} \frac{\sqrt{2} \left(\sqrt{\pi} \sqrt{2} \operatorname{erf}\left(\frac{1}{2} \sqrt{2} \rho\right) - 2 \rho e^{-\frac{1}{2} \rho^2} \right)}{\sqrt{\pi}} \quad (10)$$

> w04 := simplify(4*Pi*int(rho^2*p04*gauss,rho));

$$w04 := \frac{1}{2} \frac{\sqrt{2} \left(\rho^3 e^{-\frac{1}{2} \rho^2} + \sqrt{\pi} \sqrt{2} \operatorname{erf}\left(\frac{1}{2} \sqrt{2} \rho\right) - 2 \rho e^{-\frac{1}{2} \rho^2} \right)}{\sqrt{\pi}} \quad (11)$$

> w06 := simplify(4*Pi*int(rho^2*p06*gauss,rho));

$$w06 := \frac{1}{8} \frac{\sqrt{2} \left(-\rho^5 e^{-\frac{1}{2} \rho^2} + 9 \rho^3 e^{-\frac{1}{2} \rho^2} - 8 \rho e^{-\frac{1}{2} \rho^2} + 4 \sqrt{\pi} \sqrt{2} \operatorname{erf}\left(\frac{1}{2} \sqrt{2} \rho\right) \right)}{\sqrt{\pi}} \quad (12)$$

> w08 := simplify(4*Pi*int(rho^2*p08*gauss,rho));

$$w08 := \frac{1}{48} \frac{1}{\sqrt{\pi}} \left(\sqrt{2} \left(\rho^7 e^{-\frac{1}{2} \rho^2} - 20 \rho^5 e^{-\frac{1}{2} \rho^2} + 89 \rho^3 e^{-\frac{1}{2} \rho^2} \right) \right) \quad (13)$$

$$+ 24 \sqrt{\pi} \sqrt{2} \operatorname{erf}\left(\frac{1}{2} \sqrt{2} \rho\right) - 48 \rho e^{-\frac{1}{2} \rho^2} \Bigg) \Bigg)$$

> w10 := simplify(4*Pi*int(rho^2*p10*gauss,rho));

$$w10 := \frac{1}{384} \frac{1}{\sqrt{\pi}} \left(\sqrt{2} \left(-\rho^9 e^{-\frac{1}{2} \rho^2} + 35 \rho^7 e^{-\frac{1}{2} \rho^2} - 349 \rho^5 e^{-\frac{1}{2} \rho^2} + 1027 \rho^3 e^{-\frac{1}{2} \rho^2} \right. \right. \\ \left. \left. - 384 \rho e^{-\frac{1}{2} \rho^2} + 192 \sqrt{\pi} \sqrt{2} \operatorname{erf}\left(\frac{1}{2} \sqrt{2} \rho\right) \right) \right) \quad (14)$$

- Calculating the integration kernels by [1] Eq. (22)

> G02 := simplify(-int(w02/rho^2,rho)/(4*Pi));

$$G02 := \frac{1}{4} \frac{\operatorname{erf}\left(\frac{1}{2} \sqrt{2} \rho\right)}{\rho \pi} \quad (15)$$

> G04 := simplify(-int(w04/rho^2,rho)/(4*Pi));

$$G04 := \frac{1}{8} \frac{\sqrt{2} e^{-\frac{1}{2} \rho^2} \rho + 2 \operatorname{erf}\left(\frac{1}{2} \sqrt{2} \rho\right) \sqrt{\pi}}{\pi^{3/2} \rho} \quad (16)$$

> G06 := simplify(-int(w06/rho^2,rho)/(4*Pi));

$$G06 := \frac{1}{32} \frac{-\sqrt{2} e^{-\frac{1}{2} \rho^2} \rho^3 + 7 \sqrt{2} e^{-\frac{1}{2} \rho^2} \rho + 8 \operatorname{erf}\left(\frac{1}{2} \sqrt{2} \rho\right) \sqrt{\pi}}{\pi^{3/2} \rho} \quad (17)$$

> G08 := simplify(-int(w08/rho^2,rho)/(4*Pi));

$$G08 := \frac{1}{192} \frac{\sqrt{2} \rho^5 e^{-\frac{1}{2} \rho^2} - 16 \sqrt{2} e^{-\frac{1}{2} \rho^2} \rho^3 + 57 \sqrt{2} e^{-\frac{1}{2} \rho^2} \rho + 48 \operatorname{erf}\left(\frac{1}{2} \sqrt{2} \rho\right) \sqrt{\pi}}{\pi^{3/2} \rho} \quad (18)$$

> G10 := simplify(-int(w10/rho^2,rho)/(4*Pi));

$$G10 := \frac{1}{1536} \frac{1}{\pi^{3/2} \rho} \left(\left(-\sqrt{2} \rho^7 + 29 \sqrt{2} \rho^5 - 233 \sqrt{2} \rho^3 \right. \right. \\ \left. \left. + 384 \operatorname{erf}\left(\frac{1}{2} \sqrt{2} \rho\right) \sqrt{\pi} e^{\frac{1}{2} \rho^2} + 561 \sqrt{2} \rho \right) e^{-\frac{1}{2} \rho^2} \right) \quad (19)$$

- Calculating the velocity kernels by $K = \nabla G \times$

> K02 := simplify(diff(G02,rho));

$$K02 := -\frac{1}{4} \frac{-\sqrt{2} e^{-\frac{1}{2} \rho^2} \rho + \operatorname{erf}\left(\frac{1}{2} \sqrt{2} \rho\right) \sqrt{\pi}}{\pi^{3/2} \rho^2} \quad (20)$$

> K04 := simplify(diff(G04,rho));

$$K04 := -\frac{1}{8} \frac{\sqrt{2} e^{-\frac{1}{2} \rho^2} \rho^3 - 2 \sqrt{2} e^{-\frac{1}{2} \rho^2} \rho + 2 \operatorname{erf}\left(\frac{1}{2} \sqrt{2} \rho\right) \sqrt{\pi}}{\pi^{3/2} \rho^2} \quad (21)$$

> K06 := simplify(diff(G06,rho));

K06:= (22)

$$-\frac{1}{32} \frac{-\sqrt{2} \rho^5 e^{-\frac{1}{2} \rho^2} + 9 \sqrt{2} e^{-\frac{1}{2} \rho^2} \rho^3 - 8 \sqrt{2} e^{-\frac{1}{2} \rho^2} \rho + 8 \operatorname{erf}\left(\frac{1}{2} \sqrt{2} \rho\right) \sqrt{\pi}}{\pi^{3/2} \rho^2}$$

> K08 := simplify(diff(G08,rho));

$$K08 := -\frac{1}{192} \frac{1}{\pi^{3/2} \rho^2} \left(e^{-\frac{1}{2} \rho^2} \sqrt{2} \rho^7 - 20 \sqrt{2} \rho^5 e^{-\frac{1}{2} \rho^2} + 89 \sqrt{2} e^{-\frac{1}{2} \rho^2} \rho^3 - 48 \sqrt{2} e^{-\frac{1}{2} \rho^2} \rho + 48 \operatorname{erf}\left(\frac{1}{2} \sqrt{2} \rho\right) \sqrt{\pi} \right) \quad (23)$$

> K10 := simplify(diff(G10,rho));

$$K10 := -\frac{1}{1536} \frac{1}{\pi^{3/2} \rho^2} \left(e^{-\frac{1}{2} \rho^2} \left(-\sqrt{2} \rho^9 + 35 \sqrt{2} \rho^7 - 349 \sqrt{2} \rho^5 - 384 \sqrt{2} \rho + 1027 \sqrt{2} \rho^3 + 384 \operatorname{erf}\left(\frac{1}{2} \sqrt{2} \rho\right) \sqrt{\pi} e^{\frac{1}{2} \rho^2} \right) \right) \quad (24)$$

- Plotting the smoothing functions ζ (one-sided) [1] Eq. (23)

> p1 := plot(p02*gauss, rho = 0 .. 5, color = blue);


> p2 := plot(p04*gauss, rho = 0 .. 5, color = red);

> p3 := plot(p06*gauss, rho = 0 .. 5, color = green);

> p4 := plot(p08*gauss, rho = 0 .. 5, color = magenta);


> p5 := plot(p10*gauss, rho = 0 .. 5, color = cyan);

> display({p1,p2,p3,p4,p5});



- Plotting the G kernels (one sided, black = non-regularised kernel)

```
> p1 := plot(1/(4*Pi*x), x = 0 .. 5, y = 0 .. 0.4, color = black):
> p2 := plot(G02, rho = 0 .. 5, color = blue):
> p3 := plot(G04, rho = 0 .. 5, color = red):
> p4 := plot(G06, rho = 0 .. 5, color = green):
> p5 := plot(G08, rho = 0 .. 5, color = magenta):
> p6 := plot(G10, rho = 0 .. 5, color = cyan):
> display({p1,p2,p3,p4,p5,p6});
```



- Plotting the K kernels (one-sided, black = non-regularised kernel)

```
> p1 := plot(x/(4*Pi*x^3), x = 0 .. 5, y = 0 .. 0.5 ,color = black)
:
> p2 := plot(-K02, rho = 0 .. 5, color = blue):
> p3 := plot(-K04, rho = 0 .. 5, color = red):
> p4 := plot(-K06, rho = 0 .. 5, color = green):
> p5 := plot(-K08, rho = 0 .. 5, color = magenta):
> p6 := plot(-K10, rho = 0 .. 5, color = cyan):
> display({p1,p2,p3,p4,p5,p6});
```

